Objective Caml:

the language and its type system

Xavier Leroy INRIA Rocquencourt, France

http://caml.inria.fr/

1

Outline • Part 1: the core language (and its type system) • Part 2: objects and classes (and their type system) • Part 3: the module system (and its type system) Thesis: strong typing with type inference gives coherence to an otherwise fairly diverse language (functional and imperative and object-oriented and . . . )

The core Caml language Caml programs can be written in files and compiled separately as usual.

For the examples, we use the interactive Caml system:

~$ ocaml
        Objective Caml version 2.00
# 1+2 ;;
     - : int = 3

The system computes the type and the value of each expression entered.

3

Caml as a desktop calculator

1 + 2 * 3;;
     - : int = 7
let pi = 4.0 *. atan 1.0;;
     val pi : float = 3.14159265359
let square x = x *. x;;
     val square : float -> float = <fun>
square(sin pi) +. square(cos pi);;
     - : float = 1

let gives a name to the result of a calculation.

With one or several parameters, let defines a function.

Strong static typing with type inference

The system checks each phrase for type-correctness.

Almost no type declarations need to be provided by the user: the system infers unspecified types from the context.

Ex: in let square x = x *. x, x must have type float because it is used as argument to the float multiplication *.

Integers and floats are distinct types, with distinct operators: + and * operate on integers; +. and *. on floats.

1.0 * 2;;
     This expression has type float but is here used with type int

5

Recursive functions

Recursive functions are defined with let rec instead of let:

let rec fib n =
  if n < 2 then 1 else fib(n-1) + fib(n-2);;
     val fib : int -> int = <fun>

fib 10;;
     - : int = 89

Due to strict static scoping, the system cannot guess whether its a let or a let rec

let f x = x + 1 let f x = x + 1

let f y = f(y * 2) let rec f y = f(y * 2)

Basic data types

As usual: int, float, booleans, characters, character strings.

(1 < 2) = false;;
     - : bool = false

’a’;;
     - : char = ’a’

"Hello world";;
     - : string = "Hello world"

7

Data structures, 1: lists

Lists can be given in extension [ a; b; c ] or built from the empty list [] (nil) by adding elements in front using :: (cons).

let l = ["è"; "vero"; "è"; "ben"; "trovato"];;
     val l : string list = ["è"; "vero"; "è"; "ben"; "trovato"]

"Si" :: "non" :: l;;
     - : string list = ["Si"; "non"; "è"; "vero"; "è"; "ben"; "tro-
vato"]

The type of a list is τ list where τ is the type of its elements (lists must be homogeneous).

No explicit allocation or deallocation of “cons” cells (garbage collection). No explicit handling of pointers (done by the compiler). 8

Pattern-matching on lists

Pattern-matching is Caml’s main mechanism for inspecting and destructuring data structures.

List patterns have the exact same shape as list expressions, with identifier representing unspecified parts of the list:

• [] an empty list

• a :: b a non-empty list, with a bound to its first element

and b its tail

• [a; b; c] a 3-element list, with a, b, c bound to those

elements. 9

Insertion sort on lists

let rec sort lst =
  match lst with
    [] -> []
  | head :: tail -> insert head (sort tail)

and insert elt lst =
  match lst with
    [] -> [elt]
  | head :: tail ->
      if elt <= head then elt :: lst else head :: insert elt tail
;;
     val sort : ’a list -> ’a list = <fun>
     val insert : ’a -> ’a list -> ’a list = <fun>

sort l;;
     - : string list = ["ben"; "trovato"; "vero"; "è"; "è"]

10

Polymorphism

Notice the type inferred for sort:

val sort : ’a list -> ’a list

It means: “sort can be applied to lists of any type, and returns a list of the same type”. The type ’a is a type variable, and can be replaced by any given type.

sort [6; 2; 5; 3];;
    - : int list = [2; 3; 5; 6]

sort [3.14; 2.718];;
    - : float list = [2.718; 3.14]

We say that sort is polymorphic, and the reason it is polymorphic is because the comparison operators =, <=, etc. are themselves polymorphic: they operate between any two values of the same type. 11

Functions as values

Caml is a functional language: functions in the full mathematical sense are supported and can be passed around freely just as any other piece of data.

Example: approximating the derivative of a function.

let deriv f dx = function x -> (f(x +. dx) -. f(x)) /. dx;;
     val deriv : (float -> float) -> float -> float -> float = <fun>

let cos’ = deriv sin 1e-6;;
     val cos’ : float -> float = <fun>

cos’ pi;;
     - : float = -1.00000000014

Functions taking or returning other functions are called functionals, or higher-order functions. 12

More functionals

Function composition:

let compose f g = function x -> f(g(x));;
     val compose : (’a -> ’b) -> (’c -> ’a) -> ’c -> ’b = <fun>

let cos2 = compose square cos;;
     val cos2 : float -> float = <fun>

Generic operations over data structures (iterators):

let rec map f l =
  match l with
     [] -> []
   | hd :: tl -> f hd :: map f tl;;
     val map : (’a -> ’b) -> ’a list -> ’b list = <fun>

map (function n -> n * 2 + 1) [0; 1; 2; 3; 4];;
     - : int list = [1; 3; 5; 7; 9]

13

User-defined data structures, 1: records

The type declaration lets the user define record types and variant (disjoint sum) types.

type ratio = { num: int; denum: int };;
     type ratio = { num: int; denum: int }

let add_ratio r1 r2 =
  { num = r1.num * r2.denum + r2.num * r1.denum;
    denum = r1.denum * r2.denum }
;;
     val add_ratio : ratio -> ratio -> ratio = <fun>

14

User-defined data structures, 2: variants

The declaration of a variant type lists all possible shapes for values of that type. Each case is identified by a constructor (capitalized name). The constructor serves both for constructing values of the variant type and inspecting them by pattern-matching.

Example: symbolic computation on arithmetic expressions.

type expression =
      Const of float
    | Var of string
    | Sum of expression * expression (* e1 + e2 *)
    | Diff of expression * expression (* e1 - e2 *)
    | Prod of expression * expression (* e1 * e2 *)
    | Quot of expression * expression (* e1 / e2 *)

15

Evaluation of an expression

let rec eval env exp =
    match exp with
      Const c -> c
    | Var v -> List.assoc v env
    | Sum(f, g) -> eval env f +. eval env g
    | Diff(f, g) -> eval env f -. eval env g
    | Prod(f, g) -> eval env f *. eval env g
    | Quot(f, g) -> eval env f /. eval env g;;

     val eval : (string * float) list -> expression -> float = <fun>

env is an association list (list of pairs) mapping variable names

to their values.

eval [("x", 1.0); ("y", 3.14)] (Prod(Sum(Var "x", Const 2.0), Var "y"));;
- : float = 9.42

16

Symbolic derivation of an expression

let rec deriv exp dv =
  match exp with
    Const c -> Const 0.0
  | Var v -> if v = dv then Const 1.0 else Const 0.0
  | Sum(f, g) -> Sum(deriv f dv, deriv g dv)
  | Diff(f, g) -> Diff(deriv f dv, deriv g dv)
  | Prod(f, g) -> Sum(Prod(f, deriv g dv), Prod(deriv f dv, g))
  | Quot(f, g) -> Quot(Diff(Prod(deriv f dv, g), Prod(f, deriv g dv)),
                       Prod(g, g))
;;
     val deriv : expression -> string -> expression = <fun>

deriv (Quot(Const 1.0, Var "x")) "x";;
     - : expression =
        Quot
         (Diff (Prod (Const 0, Var "x"), Prod (Const 1, Const 1)),
          Prod (Var "x", Var "x"))

17

Imperative features

All examples so far are purely applicative (no in-place modification on data structures and no assignment to identifiers).

Caml is also equipped with full imperative power:

• Mutable data structures (modifiable in place by assignment).

• Sequence a; b

• Loops (for, while).

• Exceptions.

• Input/output by side effects. 18

Mutable data structures

Most Caml data structures are immutable (lists, sum types), but some are mutable (modifiable in place by assignment):

• Arrays τ array

• References (indirection cells; one-element arrays) τ ref

• Records declared with mutable fields

type point = { mutable x: float; mutable y: float };;

let translate p dx dy =
    p.x <- p.x +. dx; p.y <- p.y +. dy;;

19

Imperative programming: insertion sort on arrays

let insertion_sort a =
  for i = 1 to Array.length a - 1 do
    let val_i = a.(i) in
    let j = ref i in
    while !j > 0 && val_i < a.(!j - 1) do
      a.(!j) <- a.(!j - 1);
      j := !j - 1
    done;
    a.(!j) <- val_i
  done;;
     val insertion_sort : ’a array -> unit = <fun>

let a = [| 3.14; 1.2; 2.718 |];;
     val a : float array = [|3.14; 1.2; 2.718|]
insertion_sort a;;
     - : unit = ()
a;;
     - : float array = [|1.2; 2.718; 3.14|]

20

A simplified language

Expressions: a ::= c constants | x identifiers

| fun x → a function abstraction | a1 a2 function application | let x = a1 in a2 the let binding

Constants c include base constants (0, 1, false, true, "hello") and predefined operations (+, . . . ). Many derived forms, e.g. let f x y = a is

let f = fun x → fun y → a in rest of session

1

The type algebra

Type expressions: τ ::= int integers

| bool booleans | . . . other base types

| α type variables

| τ1 → τ2 function types

2

The typing rules Define the typing predicate E ⊢ a : τ (read: under assumptions E, expression a has type τ) E maps identifiers to their types.

The predicate is defined by a set of inference rules of the form P1 . . . Pn P (read: if the assumptions P1, . . . , Pn hold, then P holds)

3

Typing rules for the simply-typed λ-calculus E ⊢ x : E(x) E ⊢ c : T C(c) E ⊕ {x : τ1} ⊢ a : τ2 E ⊢ fun x → a : τ1 → τ2 E ⊢ a1 : τ ′ → τ E ⊢ a2 : τ ′ E ⊢ a1 a2 : τ E ⊢ a1 : τ ′ E ⊕ {x : τ ′} ⊢ a2 : τ E ⊢ let x = a1 in a2 : τ

T C assigns types to constants, e.g. T C(0) = int, T C(+) = int → (int → int). 4

Typing derivations

A typing derivation is a justification (a proof) of well-typing.

{x : int} ⊢ + : T C(+) {x : int} ⊢ 1 : int

{x : int} ⊢ + x : int → int {x : int} ⊢ 1 : int {x : int} ⊢ + x 1 : int ∅ ⊢ (fun x → + x 1) : int → int

5

Does this type system make sense?

A type system is sound if (compile-time) typings are consistent with the (run-time) evaluation of the program:

• A well-typed program must never go wrong at run-time by evaluating a nonsensical expression such as 1 2 or + (fun x → a).

• The final value of an expression must belong to the type assigned to the expression. E.g. if E ⊢ a : int, then a must evaluate to an integer value.

We need to specify the evaluation of an expression. 6

Reduction semantics a ∗→ a1 ∗→ a2 ∗→ . . . ∗→ v We keep rewriting the expression itself according to some evaluation rules until one of the following happens:

• We reach a normal form (an expression that cannot be reduced any further) and that normal form is a value v: the expression successfully evaluated to v (OK)

• We have an infinite reduction sequence:

the computation does not terminate (OK)

• We reach a stuck expression (a normal form that is not a value): the evaluation went wrong (bad) 7

The reduction rules Values: v ::= c | fun x → a Reduction rules: (fun x → a) v^∗→ a{x ← v} let x = v in a ∗→ b{x ← v} + n1 n2^∗→ n where n = n1 + n2 E[a]^∗→ E[b] if a^∗→ b Reduction contexts E determine the order of evaluations (here, left-to-right call-by-value):

E ::= [ ] reduction at head

| E a in function part of an application | v E in argument part of a value application | let x = E in a in left part of a let 8

Soundness of typing w.r.t. evaluation Reduction preserves typing: (subject reduction) if ∅ ⊢ a : τ and a^∗→ b, then ∅ ⊢ b : τ. Well-typed expressions are never stuck: if ∅ ⊢ a : τ and a ̸^ ∗→, then a is a value.

Corollary: soundness of typing if ∅ ⊢ a : τ, then a either reduces to a value v or loops, but never gets stuck.

9

Type inference The typing rules allow to check the validity of a given type derivation.

What we need now is an algorithm that takes a program as input and checks that it is well-typed and infers the types of its expressions.

We view this as a constraint satisfaction problem:

• Associate a type unknown Ta to each subexpression a of the program and a type unknown Tx to each identifier x.

• Based on the typing rules, derive equality constraints between the Ta and Tx.

• Find a principal solution to the constraints. 10

Setting up the constraints

• If a = x: add Ta = Tx

• If a = c: add Ta = T C(c)

• If a = fun x → b: add Ta = Tx → T_b

• If a = b(c): add T_b = Tc → Ta

• If a = let x = b in c: add Tx = T_b and Ta = Tc

The constraints C capture all possible typings of the program in the following sense: {x : Tx} ⊢ a : Ta is derivable iff the Tx, Ta are solutions of C. 11

Solving the constraints The constraints are sets of equations between first-order terms (where the Tx and Ta are viewed as variables). These constraints are solved by the standard unification algorithm between first-order terms.

Given a set C of constraints, the algorithm either

• fails, meaning that C has no solutions;

• or returns a subtitution φ assigning types to the Tx and Ta that is a principal solution of C:

– (correctness) φ is a solution of C, i.e. for all equations τ1 = τ2 in C, we have φ(τ1) = φ(τ2) – (principality) if ψ is a solution if C, there exists θ such that ψ = θ ∘ φ. 12

An example of type inference Consider a = fun x → fun y → 1 and b = a true. We obtain the following equations: Ta = Tx → Ty → T1 T1 = int Ta = T_{true} → T_b T_{true} = bool A principal solution is:

Tx = bool Ty = α Ta = bool → α → int T_b = α → int All other solutions are obtained by replacing α by any type expression τ.

13

In summary Combining constraint building and constraint solving, we obtain an algorithm W (a) which either

• fails, meaning that a is ill-typed

(there does not exist τ such that ∅ ⊢ a : τ) • or returns a type τ which is the principal type of a: – ∅ ⊢ a : τ – if ∅ ⊢ a : τ^′, then τ^′ = θ(τ) for some substitution θ.

We say that the type system admits principal types. 14

Adding polymorphism

The current type system admits undetermined types (α) but not polymorphism: there is no way to use a function with several different types.

let f = fun x → x in

at this point, f : α → α

let y = f 1 in

here we must take α = int

f true

we get at type error as int and bool are incompatible

15

Type schemes

Type schemes: σ ::= ∀α1, . . . , αn. τ

Type schemes are a compact representation of all possible types of a polymorphic expression.

Example: ∀α. α → α stands for the set of types {τ → τ | τ type}, which are all acceptable types for fun x → x.

Instanciation relation: τ ≤ σ (read τ is an instance of σ) if σ = ∀α1, . . . , αn. τ^′ and τ = τ^′{α1 ← τ1, . . . , αn ← τn} for suitable types τ1 . . . τn.

16

The polymorphic let rule We allow let-bound identifiers to be assigned type schemes in the typing environment. τ ≤ E(x) E ⊢ x : τ E ⊢ a : τ1 E ⊕ {x : Gen(τ1, E)} ⊢ b : τ E ⊢ let x = a in b : τ Gen(τ, E) = ∀α1, . . . , αn. τ where {α1, . . . , αn} = F V (τ) \ F V (E). (The α_i are the variables that occur in τ and that are generalizable at this point because they are completely unconstrained, i.e. they do not occur elsewhere in the typing environment E.) 17

Example of polymorphic typing

{x : α} ⊢ x : α ∅ ⊢ fun x → x : α → α int → int ≤ ∀α.α → α {f : ∀α.α → α} ⊢ f : int → int {f : ∀α.α → α} ⊢ 1 : int {f : ∀α.α → α} ⊢ f 1 : int ∅ ⊢ let f = fun x → x in f 1 : int

18

Results The previous results extend to the system with

let-polymorphism:

• Type soundness (subject reduction, etc)

• Existence of principal typings; the type inference algorithm.

(Although now we need to interleave constraint building and constraint solving, so that Gen at let points is handled correctly.)

(See course handout for detailed proofs and algorithms.)

19

Extension 1: recursion

a ::= . . . | let rec x = a in b Typing rule: E ⊕ {x : τ1} ⊢ a : τ1 E ⊕ {x : Gen(τ1, E)} ⊢ b : τ E ⊢ let rec x = a in b : τ

Notice that x must be used monomorphically in a; it is polymophic only in b.

let rec f x = f 1; x in f true;;
     This expression has type bool but is here used with type int

f has type int → int instead of ∀α. α → α (which would be

semantically correct) because f was used with type int → int in the recursion. 20

Polymorphic recursion We can allow polymorphic uses of a recursively-bound identifier by “guessing” a type scheme for it rather than a simple type:

E ⊕ {x : σ} ⊢ a : τ1 σ = Gen(τ1, E) E ⊕ {x : σ} ⊢ b : τ E ⊢ let rec x = a in b : τ

However, this renders type inference undecidable (type inference then reduces to a semi-unification problem instead of a unification problem, and semi-unification is undecidable).

21

Extension 2: data types and pattern-matching We treat constructors as new constants with associated type schemes. For instance:

type ’a list = Nil | Cons of ’a * ’a list

T C(Nil) = ∀α. α list T C(Cons) = ∀α. α × α list → α list We add pattern-matching:

Expressions: a ::= . . . | match a with p1 → a1 | pn → an Patterns: p ::= | x | c | c(p1, . . . , pn)

Typing match and patterns E ⊢ a : τ ′ ⊢ pi : τ ′ ⇒ Ei E ⊕ Ei ⊢ ai : τ for all i = 1 . . . n E ⊢ match a with p1 → a1 | pn → an : τ ⊢ p : τ ⇒ E means “pattern p matches values of type τ and binds identifiers as described in E”. ⊢ : τ ⇒ ∅ ⊢ x : τ ⇒ {x : τ} τ ≤ T C(c) ⊢ c : τ ⇒ ∅ τ1 × · · · × τn → τ ≤ T C(c) for all i = 1 . . . n, ⊢ pi : τi ⇒ Ei ⊢ c(p1, . . . , pn) : τ ⇒ E1 ⊕ · · · ⊕ En 23

Extension 3: references and assignment We model mutable data structures with references (indirection cells).

τ ref is the type of references containing a value of type τ.

Operations:

• Create a new reference

ref : ∀α. α → α ref • Dereferencing (access the contents of a reference) ! : ∀α. α ref → α • Assignment (change the contents of a reference)

:= : ∀α. α ref → α → unit

The problem with polymorphic references

When references are added, the monomorphic type system remains sound, but the polymorphic type system becomes unsound:

let r = ref(fun x → x) in

here, r : ∀α. (α → α) ref

r := (fun x → x + 1);

we use r with type (int→int) ref; fine

(!r)(true)

we use r with type (bool→bool) ref; fine

The program gets stuck at run-time trying to evaluate true + 1.

25

The value restriction

The simplest and most practical solution to the problem with polymorphic references is to restrict let-generalization to expressions that are syntactically values (i.e. constants and function abstractions):

E ⊢ a : τ1 σ = (_{Gen}(_{τ1, E}) if_a is a value;

τ1 otherwise E ⊕ {x : σ} ⊢ b : τ E ⊢ let x = a in b : τ

This rules out the problem, since ref a is not a value.

The value restriction in practice

let f = fun x → y → y;;
                            f : ∀α, β. α → β → β
let g = f 1;;
                            g : γ → γ
g 1;;
                            - : int
g;;
                            - : int → int
g true;;
                            Type error
let g = fun y → f 1 y;;
                            g : ∀γ. γ → γ
g 1;;
                            - : int
g true;;
                            - : bool

Works well in practice because 99% of all useful polymorphic values are functions fun x → a.